# Rules for integrands of the form $(d\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n$

**1.** $\int (d\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+$

**1:** $\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{x}\,dx$ when $n\in\mathbb{Z}^+$

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{x} = \text{Subst}\left[\frac{1}{\text{Tanh}[x]},\,x,\,\text{ArcSinh}[c\,x]\right]\,\partial_x\text{ArcSinh}[c\,x]$
- **Note:** $\frac{(a+b\,x)^n}{\text{Tanh}[x]}$ is not integrable unless $n\in\mathbb{Z}^+$.
- **Rule: If** $n\in\mathbb{Z}^+$, **then**

$$\int \frac{(a+b\,\text{ArcSinh}[c\,x])^n}{x}\,dx\;\longrightarrow\;\text{Subst}\left[\int \frac{(a+b\,x)^n}{\text{Tanh}[x]}\,dx,\,x,\,\text{ArcSinh}[c\,x]\right]$$

- **Program code:**

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./x_,x_Symbol] :=
  Subst[Int[(a+b*x)^n/Tanh[x],x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./x_,x_Symbol] :=
  Subst[Int[(a+b*x)^n/Coth[x],x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[n,0]
```

**2:** $\int (d\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n\in\mathbb{Z}^+\wedge m\neq -1$

- **Derivation: Integration by parts**
- **Rule: If** $n\in\mathbb{Z}^+\wedge m\neq -1$, **then**

$$\int (d\,x)^m\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx\;\longrightarrow\;\frac{(d\,x)^{m+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{d\,(m+1)}-\frac{b\,c\,n}{d\,(m+1)}\int \frac{(d\,x)^{m+1}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}}{\sqrt{1+c^2\,x^2}}\,dx$$

- **Program code:**

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcSinh[c*x])^n/(d*(m+1)) -
  b*c*n/(d*(m+1))*Int[(d*x)^(m+1)*(a+b*ArcSinh[c*x])^(n-1)/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcCosh[c*x])^n/(d*(m+1)) -
  b*c*n/(d*(m+1))*Int[(d*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

**2.** $\int x^m (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+$

**1:** $\int x^m (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+\wedge n>0$

**Derivation: Integration by parts**

**Rule: If $n\in\mathbb{Z}^+\wedge m\neq-1$, then**

$$\int x^m (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \frac{x^{m+1}(a+b\,\text{ArcSinh}[c\,x])^n}{m+1}-\frac{b\,c\,n}{m+1}\int\frac{x^{m+1}(a+b\,\text{ArcSinh}[c\,x])^{n-1}}{\sqrt{1+c^2x^2}}\,dx$$

**Program code:**

```
Int[x_^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  x^(m+1)*(a+b*ArcSinh[c*x])^n/(m+1) -
  b*c*n/(m+1)*Int[x^(m+1)*(a+b*ArcSinh[c*x])^(n-1)/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GtQ[n,0]
```

```
Int[x_^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  x^(m+1)*(a+b*ArcCosh[c*x])^n/(m+1) -
  b*c*n/(m+1)*Int[x^(m+1)*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GtQ[n,0]
```

2. $\int x^m (a + b \operatorname{ArcSinh}[c x])^n \, dx$ when $m \in \mathbb{Z}^+ \land n < -1$

1: $\int x^m (a + b \operatorname{ArcSinh}[c x])^n \, dx$ when $m \in \mathbb{Z}^+ \land -2 \le n < -1$

**Derivation: Integration by parts and integration by substitution**

- **Basis:** $\frac{(a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{1+c^2 x^2}} = \partial_x \frac{(a+b \operatorname{ArcSinh}[c x])^{n+1}}{b c (n+1)}$
- **Basis:** $\frac{F[x]}{\sqrt{1+c^2 x^2}} = \frac{1}{c} F\left[\frac{\operatorname{Sinh}[\operatorname{ArcSinh}[c x]]}{c}\right] \partial_x \operatorname{ArcSinh}[c x]$
- **Basis: If** $c > 0 \lor m \in \mathbb{Z}$, **then** $\frac{x^{m-1} (m+(m+1) c^2 x^2)}{\sqrt{1+c^2 x^2}} = \frac{1}{c^m} \operatorname{Subst}\left[\operatorname{Sinh}[x]^{m-1} \left(m + (m+1) \operatorname{Sinh}[x]^2\right), x, \operatorname{ArcSinh}[c x]\right] \partial_x \operatorname{ArcSinh}[c x]$

**Note: Although not essential, by switching to the hyperbolic trig world this rule saves numerous steps and results in more compact antiderivatives.**

- **Rule: If** $m \in \mathbb{Z}^+ \land -2 \le n < -1$, **then**

$$\int x^m (a + b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{x^m \sqrt{1 + c^2 x^2} \, (a + b \operatorname{ArcSinh}[c x])^{n+1}}{b c (n + 1)} - \frac{1}{b c (n + 1)} \int \frac{x^{m-1} \left(m + (m + 1) c^2 x^2\right) (a + b \operatorname{ArcSinh}[c x])^{n+1}}{\sqrt{1 + c^2 x^2}} \, dx \longrightarrow$$

$$\frac{x^m \sqrt{1 + c^2 x^2} \, (a + b \operatorname{ArcSinh}[c x])^{n+1}}{b c (n + 1)} - \frac{1}{b c^{m+1} (n + 1)} \operatorname{Subst}\left[\int (a + b x)^{n+1} \operatorname{Sinh}[x]^{m-1} \left(m + (m + 1) \operatorname{Sinh}[x]^2\right) dx, x, \operatorname{ArcSinh}[c x]\right]$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[1+c^2*x^2]*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  1/(b*c^(m+1)*(n+1))*Subst[Int[ExpandTrigReduce[(a+b*x)^(n+1),Sinh[x]^(m-1)*(m+(m+1)*Sinh[x]^2),x],x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GeQ[n,-2] && LtQ[n,-1]
```

```
Int[x_^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[-1+c*x]*Sqrt[1+c*x]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  1/(b*c^(m+1)*(n+1))*Subst[Int[ExpandTrigReduce[(a+b*x)^(n+1)*Cosh[x]^(m-1)*(m-(m+1)*Cosh[x]^2),x],x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GeQ[n,-2] && LtQ[n,-1]
```

**2:** $\int x^m (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $m\in\mathbb{Z}^+\wedge n<-2$

**Derivation: Integration by parts and algebraic expansion**

- **Basis:** $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2x^2}} = \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}$
- **Basis:** $\partial_x\left(x^m\sqrt{1+c^2x^2}\right) = \frac{m\,x^{m-1}}{\sqrt{1+c^2x^2}}+\frac{c^2(m+1)\,x^{m+1}}{\sqrt{1+c^2x^2}}$

**Rule: If** $m\in\mathbb{Z}^+\wedge n<-2$**, then**

$$\int x^m (a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{x^m\sqrt{1+c^2x^2}\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)} -$$

$$\frac{m}{b\,c\,(n+1)}\int\frac{x^{m-1}(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{\sqrt{1+c^2x^2}}\,dx-\frac{c\,(m+1)}{b\,(n+1)}\int\frac{x^{m+1}(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{\sqrt{1+c^2x^2}}\,dx$$

**Program code:**

```
Int[x_^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[1+c^2*x^2]*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  m/(b*c*(n+1))*Int[x^(m-1)*(a+b*ArcSinh[c*x])^(n+1)/Sqrt[1+c^2*x^2],x] -
  c*(m+1)/(b*(n+1))*Int[x^(m+1)*(a+b*ArcSinh[c*x])^(n+1)/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && LtQ[n,-2]
```

```
Int[x_^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[-1+c*x]*Sqrt[1+c*x]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  m/(b*c*(n+1))*Int[x^(m-1)*(a+b*ArcCosh[c*x])^(n+1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x] -
  c*(m+1)/(b*(n+1))*Int[x^(m+1)*(a+b*ArcCosh[c*x])^(n+1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && LtQ[n,-2]
```

**3:** $\int x^m (a + b\,\text{ArcSinh}[c\,x])^n\,dx$ when $m \in \mathbb{Z}^+$

- **Derivation: Integration by substitution**
- **Basis:** $F[x] == \frac{1}{c}\,\text{Subst}\left[F\left[\frac{\text{Sinh}[x]}{c}\right]\,\text{Cosh}[x],\,x,\,\text{ArcSinh}[c\,x]\right]\,\partial_x\,\text{ArcSinh}[c\,x]$
- **Note: If** $m \in \mathbb{Z}^+$**, then** $(a + b\,x)^n\,\text{Sinh}[x]^m\,\text{Cosh}[x]$ **is integrable in closed-form.**
- **Rule: If** $m \in \mathbb{Z}^+$**, then**

$$\int x^m (a + b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \frac{1}{c^{m+1}}\,\text{Subst}\left[\int (a + b\,x)^n\,\text{Sinh}[x]^m\,\text{Cosh}[x]\,dx,\,x,\,\text{ArcSinh}[c\,x]\right]$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  1/c^(m+1)*Subst[Int[(a+b*x)^n*Sinh[x]^m*Cosh[x],x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[m,0]
```

```
Int[x_^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  1/c^(m+1)*Subst[Int[(a+b*x)^n*Cosh[x]^m*Sinh[x],x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[m,0]
```

**X:** $\int (d\,x)^m (a + b\,\text{ArcSinh}[c\,x])^n\,dx$

- **Rule:**

$$\int (d\,x)^m (a + b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \int (d\,x)^m (a + b\,\text{ArcSinh}[c\,x])^n\,dx$$

- **Program code:**

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d*x)^m*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d*x)^m*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```